

Models for Scheduling Individual Jet Aircraft

Hongsuk Yang²

Graduate School of Business, Seoul National University, 599 Gwanangno, Gwanak-gu Seoul,
Korea 151-916, E-mail: hongsuk@snu.ac.kr

Abstract

This paper considers the short term fleet scheduling problem as described by Keskinocak and Tayur (1998). Fleet scheduling may directly affect the service quality of fractional jet aircraft business. The contributions of this paper are two: (i) we show how their model is easily implemented in a standard modeling language, LINGO, and (ii) an alternate formulation is given which is expected to perform better on large, difficult problems.

Key Words: Fleet Scheduling, Crew Scheduling, Service Quality

1. Introduction

Fleet planning and routing decisions in the airline and similar industries can be thought of as being of three types: (a) long range planning, e.g., fleet sizing and route structure, (b) medium term, e.g., given a route structure and fleet composition, assign vehicle types to the routes, and (c) short term, e.g., which specific vehicles should be assigned to each route in the next few days, given maintenance requirements of individual aircraft, unforeseen aircraft unavailabilities because of weather, etc.

For example, regarding (a), Icelandair was one of the first to choose a hub-and-spoke route structure to provide cheap fare to trans-Atlantic passengers, using Reykjavik as its hub linking European cities to eastern cities of the US (Passell, 1994). Regarding (b), American Airlines is trying to change fleets from turboprops to regional jets (Ziemba and Washburn, 1997). Delta Airlines has chosen to use at least 10 different vehicle types, e.g., L-1011, B-737, etc. to service its approximately 2500 domestic flight legs per day.

Some discussions of the fleet assignment problem at American Airlines and Delta Airlines can be found in Abara (1989) and Subramanian *et al.* (1994). With regard to (c), Keskinocak and Tayur (1998) developed a mathematical model for scheduling individual jet aircraft for a "jet taxi" service. Gershkoff (1989) and Anbil *et al.* (1991), discuss methods for crew scheduling to individual flights. Fleet scheduling may directly affect the service quality of fractional jet aircraft business. Customers of the fractional jet business never expect the un-

† Corresponding Author

availability of aircraft for the requested trips.

This paper considers the short term fleet scheduling problem as described by Keskinocak and Tayur (1998). The contributions of this paper are two: (i) we show how their model is easily implemented in a standard modeling language, LINGO, and (ii) an alternate formulation is given which is expected to perform better on large, difficult problems.

2. Example

We first illustrate the model of Keskinocak and Tayur. The example has 4 aircrafts, and 8 trips to be covered, involving 10 cities. The scheduling horizon is 800 minutes. Thus, only aircraft 1 is maintenance constrained. Any trip that cannot be covered by one of our four owned aircraft must be covered by an expensive rental aircraft. The data for the example are summarized in the Table 1, Table 2 and Table 3.

Table 1. Information about the aircraft in Example

Aircraft	Initial Location	Max Flying Hrs	Max Landings	Max Time
1	6	337	9	630
2	7	800	80	800
3	2	800	80	800
4	4	800	80	800

Table 2. Information about the trips in Example

Trip	Departure location	Destination	Departure time	Flight hours	Total hours	No. of landings	Scheduled aircraft
1	2	2	210	220	250	1	3
2	3	7	650	90	120	1	2
3	9	5	298	120	150	1	0
4	6	8	35	150	180	1	0
5	8	5	293	258	288	1	0
6	8	10	385	141	411	2	0
7	4	1	14	201	231	1	0
8	6	6	188	60	90	1	0

Table 3. Flight time for positioning legs in Example

City	1	2	3	4	5	6	7	8	9	10
1	0	150	190	201	108	95	108	124	67	134
2	150	0	277	342	258	242	190	228	175	30
3	190	277	0	150	192	212	90	67	124	247
4	201	342	150	0	120	150	175	134	170	319
5	108	258	192	120	0	30	153	134	120	242
6	95	242	212	150	30	0	162	150	124	228
7	108	190	90	175	153	162	0	42	42	162
8	124	228	67	134	134	150	42	0	60	201
9	67	175	124	170	120	124	42	60	0	150
10	134	30	247	319	242	228	162	201	150	0

3. LINGO MODEL

LINGO is a widely used optimization language for nonlinear programming. We write the example with a standard modeling language, LINGO as follows:

```

MODEL:
! Aircraft Scheduling (Aircraft);
! Aircraft routing and assignment with multiple aircrafts and subcontract;

SETS:
Aircraft/1..4/: gamma, mhour, mland, mtime, mno, Zi00;
Trip/1..8/: alpha, beta, dtime, fly, tti, land, sch, Sub;
Position/1..10/: ;
PoPo(Position, Position): ftime, landings;
AiTr(Aircraft, Trip): AT, Zi0k, Zij0;
TrTr(Trip, Trip): TT;
AiTrTr(Aircraft, Trip, Trip): Zijk;
ENDSETS

DATA:
gamma = 6 7 2 4; ! Initial location of aircraft I;
mhour = 337 800 800 800; ! Max flying hours before maintenance;
mland = 9 80 80 80; ! Max landing before maintenance;

```

```

mtime = 630 800 800 800; ! Max time until maintenance;
mno = 0 0 0 0; ! Label of trip, which is the scheduled maintenance for aircraft I;
alpha = 2 3 9 6 8 8 4 6; ! Departure location of trip J;
beta = 2 7 5 8 5 10 1 6; ! Destination for trip J;
dtime = 210 650 298 35 293 385 14 188; ! Departure time for trip J;
fly = 220 90 120 150 258 141 201 60; ! Flight time for trip J;
tti = 250 120 150 180 288 411 231 90; ! Total travel time for trip J;
land = 1 1 1 1 1 2 1 1; ! Number of landings for trip J;
sch = 3 2 0 0 0 0 0 0; ! Label of aircraft, which is scheduled to trip J;
ftime = 0 150 190 201 108 95 108 124 67 134
        150 0 277 342 258 242 190 228 175 30
        190 277 0 150 192 212 90 67 124 247
        201 342 150 0 120 150 175 134 170 319
        108 258 192 120 0 30 153 134 120 242
        95 242 212 150 30 0 162 150 124 228
        108 190 90 175 153 162 0 42 42 162
        124 228 67 134 134 150 42 0 60 201
        67 175 124 170 120 124 42 60 0 150
        134 30 247 319 242 228 162 201 150 0; ! Flight time from position X to position Y;
landings = 1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1
           1 1 1 1 1 1 1 1 1 1; ! Number of landings during the flight from position X to
position Y;
ENDDATA

! If aircraft I can serve trip J (if trip J was the only trip to be scheduled), then AT(I, J) = 1;
@FOR(AiTr(I, J): AT(I, J) =
    ((sch(J) #EQ# 0) #OR# ((SCh(J) #NE# 0) #and# (Sch(J) #EQ# I)))
#and# (ftime(gamma(I),alpha(J)) #LE# dtime(J))
#and# ((mno(I) #EQ# 0) #and# (dtime(J) + tti(J) #LE# mtime(I)));
! If trip K can be served immediately after trip J by the same aircraft, then TT(J, K) = 1;
@FOR(TrTr(J, K): TT(J, K) = (#NOT# (sch(J)*sch(K) #NE# 0))
#and# (dtime(J)+tti(J)+ftime(beta(J), alpha(K)) #LE# dtime(K)));
! For unscheduled trip J, if trip J is subcontracted, then Sub(J) = 1;
@FOR(Trip(J)|sch(J) #EQ# 0: @BIN(Sub(J)));
@FOR(Trip(J)|sch(J) #NE# 0: Sub(J) = 0);
! If aircraft I serves trip J just before trip K, then Zij = 1 ;
@FOR(AiTrTr(I, J, K)|AT(I, J)*AT(I, K)*TT(J, K) #EQ# 1: @BIN(Zijk(I, J, K)));

```

```

@FOR(AiTr(I, J, K)|AT(I, J)*AT(I, K)*TT(J, K) #NE# 1: Zijk(I, J, K) = 0);
! If aircraft I does not serve any trips, then Zi00 = 1;
@FOR(Aircraft(I): @BIN(Zi00(I)));
! If trip K is the first trip served by aircraft I, then Zi0k = 1;
@FOR(AiTr(I, K)|AT(I, K) #EQ# 1: @BIN(Zi0k(I, K)));
@FOR(AiTr(I, K)|AT(I, K) #NE# 1: Zi0k(I, K)=0);
! If trip J is the last trip served by aircraft I, then Zij0 = 1;
@FOR(AiTr(I, J)|AT(I, J) #EQ# 1: @BIN(Zij0(I, J)));
@FOR(AiTr(I, J)|AT(I, J) #NE# 1: Zij0(I, J)=0);
! Objective function:
Minimize (1) the flight hours of positioning legs for the first trip,
          (2) the flight hours of positioning legs for the connected trips, and
          (3) Constant * (the flight hours of subcontracting);
MIN = @SUM(Aircraft(I):
      @SUM(Trip(K): ftime(gamma(I), alpha(K)) * Zi0k(I, K)))
+@SUM(Aircraft(I):
      @SUM(Trip(J):
          @SUM(Trip(K): ftime(beta(J), alpha(K)) * Zijk(I,J,K))))
+10 * @SUM(Trip(J): fly(J) * Sub(J));
! Each trip, if not scheduled, must be covered;
@FOR(Trip(K)|sch(K) #EQ# 0:
      @SUM(Aircraft(I): Zi0k(I, K) + @SUM(Trip(J): Zijk(I, J, K))) + Sub(K) = 1);
! Each trip, if scheduled, must be covered;
@FOR(Trip(K)|sch(K) #NE# 0:
      @SUM(Trip(J): Zijk(sch(K), J, K) + Zi0k(sch(K), K) = 1);
! Flow out of trip J = flow into trip J for aircraft I;
@FOR(AiTr(I, J)| AT(I, J) #EQ# 1:
      @SUM(Trip(K): Zijk(I, J, K) + Zij0(I, J)
      = @SUM(Trip(Q): Zijk(I, Q, J) + Zi0k(I, J));
! Each aircraft I must have a first trip K;
@FOR(Aircraft(I): @SUM(Trip(K): Zi0k(I, K) + Zi00(I) = 1);
! Max flying hours for aircraft I, with no scheduled maintenance trip;
@FOR(Aircraft(I)|mno(I) #EQ# 0:
      @SUM(Trip(K): (ftime(gamma(I),alpha(K)) + fly(K) ) * Zi0k(I, K)
      + @SUM(Trip(J):
          @SUM(Trip(K):
              (ftime(beta(J), alpha(K)) + fly(K) ) * Zijk(I, J, K)))
      <= mhour(I));
! Max flying hours for aircraft I, with scheduled maintenance trip;
@FOR(Aircraft(I)|mno(I) #NE# 0:
      @SUM(Trip(K)|TT(K, mno(I)) #EQ# 1:
          (ftime(gamma(I), alpha(K)) + fly(K)) * Zi0k(I, K)
      + @SUM(Trip(J)|TT(J, mno(I)) #EQ# 1:
          @SUM(Trip(K)|TT(K, mno(I)) #EQ# 1:

```

```

                (ftime(beta(J),alpha(K)) + fly(K)) * Zijk(I, J, K))
    <= mhour(I);
! Max landings for aircraft I, with no scheduled maintenance trip;
@FOR(Aircraft(I)|mno(I) #EQ# 0:
    @SUM(Trip(K):
        (landings(gamma(I), alpha(K)) + land(K)) * Zi0k(I, K))
    + @SUM(Trip(J):
        @SUM(Trip(K):
            (landings(beta(J), alpha(K)) + land(K)) * Zijk(I, J, K)))
    <= mland(I);
! Max landings for aircraft I, with scheduled maintenance trip;
@FOR(Aircraft(I)|mno(I) #NE# 0:
    @SUM(Trip(K)|TT(K, mno(I)) #EQ# 1:
        (landings(gamma(I), alpha(K)) + land(K)) * Zi0k(I, K))
    + @SUM(Trip(J)|TT(J, mno(I)) #EQ# 1:
        @SUM(Trip(K)|TT(K, mno(I)) #EQ# 1:
            (landings(beta(J), alpha(K)) + land(K)) * Zijk(I, J, K)))
    <= mland(I);
END

```

4. Results

The decision variables in the model are Sub's, Zi00's, Zi0k's, Zij0's, and Zijk's. To save space, we restrict the model to show only nonzero variables.

Objective value:	3138.000	
Variable	Value	Reduced Cost
SUB(5)	1.000000	-2580.000
ZI0K(1, 4)	1.000000	0.000000
ZI0K(2, 8)	1.000000	162.0000
ZI0K(3, 1)	1.000000	0.000000
ZI0K(4, 7)	1.000000	0.000000
ZIJ0(1, 3)	1.000000	0.000000
ZIJ0(2, 2)	1.000000	0.000000
ZIJ0(3, 1)	1.000000	0.000000
ZIJ0(4, 6)	1.000000	0.000000
ZIJK(1, 4, 3)	1.000000	60.00000
ZIJK(2, 8, 2)	1.000000	212.0000
ZIJK(4, 7, 6)	1.000000	124.0000

The LINGO solutions are interpreted and summarized as follows:

Table 4. Results of the Example

Aircraft	Trips
1	3, 4
2	2, 8
3	1
4	6, 7
Subcontract	5

The fractional jet business in the case is assumed to have own four jet aircrafts. The aircraft 1 with limited availability covers trip 3 and 4. The aircraft 2 covers trip 2 and 8. The aircraft 3 covers trip 1. The aircraft 4 covers trip 6 and 7. With four aircrafts, they are not able to cover the whole 8 trips. Therefore, the trip 5 should be covered by a subcontract.

5. Alternative Model

An alternative approach to this problem is based on the method that is widely used for crew scheduling. The basic idea is to generate beforehand all possible itineraries (or tours) for each aircraft. For our example with 8 trips to be covered there are at most $2^8 = 256$ tours for each aircraft. Because of sequencing and other restrictions, there are in fact only 14 tours total for the four owned aircraft and six (trivial) tours for the possible rentals. We want to solve the integer program:

Minimize the cost of the tours selected;

Subject to:

For each trip:

On of the selected tours must include this trip;

For each aircraft:

At most one tour can be selected for it;

This model may have many more variables than the first model, but much fewer constraints. We expect it to be easier to solve for big problems because the “tour construction” component of the problem has been pre-solved. The results of model for the example problem are as follows:

The interpreted results are the same as in the previous section.

Total Cost

3138

Cost	124	0	60	90	234	374	0	170	134	134	0	150	124	1200	1500	2580	1410	2010	600	
Schedule	0	0	0	1	0	0	1	1	0	0	0	0	0	1	0	0	1	0	0	0

Trip1																					1	=	1	
Trip2					1	1	1															1	=	1
Trip3	1			1		1			1					1								1	=	1
Trip4		1		1											1							1	=	1
Trip5									1								1					1	=	1
Trip6										1				1					1			1	=	1
Trip7											1			1							1	1	=	1
Trip8			1				1							1								1	=	1
Aircraft1	1	1	1	1																		1	<=	1
Aircraft2					1	1	1															1	<=	1
Aircraft3								1														1	<=	1
Aircraft4									1	1	1	1	1	1	1							1	<=	1

Figure 1. Result of crew scheduling model

6. Conclusion

Fractional jet service is currently popular in the United States and Latin America and gets more attention in the East Asian region as Chinese and other Asian countries' economy grows. With limited number of aircrafts, covering requested trips may be impossible without using many costly subcontracts unless the fractional jet service does not have careful optimization plan. But if the business fails to serve the requested trips, it will not only directly affect the service quality but also may threaten the business itself.

This paper deals with a short term fleet scheduling problem of fractional jet service. In this paper we show how their model is easily implemented in a standard modeling language, LINGO. With the idea of crew scheduling, we also suggested an alternate formulation which is expected to perform better on large, difficult problems.

Acknowledgement

This research was supported by The Institute of Management Research.

Reference

1. Anbil, R., E. Gelman, Patty, B., and Tanga, R.(1991), Recent Advances in Crew-pairing Optimization at American Airlines, *Interfaces*, Vol. 21, pp. 62-74.
 2. Abara, J.(1989), Applying Integer Linear Programming to the Fleet Assignment Problem, *Interfaces*, Vol. 19, No. 4, pp. 20-28.
 3. Gershkoff, I.(1989), Optimizing Flight Crew Schedules, *Interfaces*, Vol. 19, No. 4, pp. 29-43.
 4. Keskinocak, P. and Tayur, S.(1998), Scheduling of Time-shared Aircraft, *Transportation Science*, Vol. 32, No. 3, pp. 277-294.
 5. Passell, P.(1994), Filling A Trans-Atlantic Air Niche, *New York Times*.
 6. Schrage, L.(1998), *Optimization Modeling with LINGO*. LINDO Systems Inc.
 7. Subramanian, R., R. Scheff Jr., Quillinan, J., Wiper, D., and Roy Marsten(1994), Coldsart: Fleet Assignment at Delta Air Lines. *Interfaces*, Vol. 24, No. 1, pp. 104-120.
 8. Ziemba, S. and Washburn, G.(1997), Small Jets The Big Issue, *Chicago Tribune*.
-